



2010

Teaching the Budgeting Process Using a Spreadsheet Template

Benoit Boyer

Sacred Heart University, boyerb@sacredheart.edu

Follow this and additional works at: http://digitalcommons.sacredheart.edu/wcob_fac



Part of the [Accounting Commons](#), and the [Finance and Financial Management Commons](#)

Recommended Citation

Boyer, Benoit, "Teaching the Budgeting Process Using a Spreadsheet Template" (2010). *WCOB Faculty Publications*. Paper 45.
http://digitalcommons.sacredheart.edu/wcob_fac/45

This Article is brought to you for free and open access by the Jack Welch College of Business at DigitalCommons@SHU. It has been accepted for inclusion in WCOB Faculty Publications by an authorized administrator of DigitalCommons@SHU. For more information, please contact ferribyp@sacredheart.edu.

Teaching the Budgeting Process Using a Spreadsheet Template

Benoît N. Boyer,
Professor of Accounting and Chair of the Accounting and Information Systems Department,
Sacred Heart University, Fairfield, CT.

Abstract

The purpose of this article is to demonstrate how to use a template to teach the budgeting process and how to use the same template to do a sensitivity analysis once the budget process is completed. The template can be used as-is by non-accounting majors. They just have to fill in all of their assumptions by replacing the red numbers. The template can be locked so that students cannot modify the formulas by mistake. Simply protect the worksheet (Review tab > Protect Sheet) and students will be able to only change the assumptions. Accounting majors are shown how to build their own template by using formulas. This requires a good understanding of Excel and of the budgeting process. Once the template is constructed, making modifications to the assumptions is very easy and students can see the results immediately.

Introduction

Most textbooks in accounting or in entrepreneurship use tables to explain budgeting. The textbooks usually explain one equation at a time and students rapidly learn that creating a budget by hand is a very long process. In explaining a business plan, many entrepreneurship books have a special business plan appendix, but most of their examples are based on the acquisition of an existing company (Kuratko, 2009). By using Excel, many professors have reduced the amount of time spent doing calculations. However, most accounting professors continue to use Excel as a calculator and do not develop a spreadsheet that would allow a rapid analysis of “what if” situations. This paper demonstrates how to create a template that allows students to change one or many of their assumptions and see the impact on the budget immediately. Unlike most examples used in books, this template also incorporates income taxes.

I developed the current template at the request of the organizers of the Connecticut Business Plan Competition: The Entrepreneurship Foundation. The competition started more than ten years ago. It involves mostly graduate students, and all colleges and universities in Connecticut can send teams to the competition. The biggest difficulty that students have had was doing a budgetary analysis. Most were enrolled in an Entrepreneurship class and were not accounting majors. They understood the concept of a budget but were unable to do the analysis themselves. The Foundation features this template in their student guide (The Entrepreneurship Foundation, 2009).

After a long discussion with the Foundation, we decided that the template would have a section for assumptions, a cash budget, an income statement, and a balance sheet. The assumptions worksheet would allow students to input their data and then see the results immediately in the budget worksheet.

Description of the Template

The first screen students see is the Assumptions Sheet.

Figure 1:
Budget Assumptions

	A	B	C	D	E	F	G	H
1	Fill the Red Numbers with your information							
2			Q1	Q2	Q3	Q4	Q5	
3	Selling price per unit	\$ 25.00						
4	Total market per year	1,000,000						
5	Market share	10%						
6	Annual Growth	5%						
7	Percentage of sales per quarter		10%	30%	40%	20%		
8	Sales per quarter		10,000	30,000	40,000	20,000		
9	Cash receipts same quarter	70%						
10	Ending inventory finished goods	25%	next quarter sales					
11	Raw material per unit	15	pounds per unit					
12	Raw material inventory	20%	next quarter purchases					
13	Cost of raw materials	\$ 0.25	per pound					
14	Cash payments same quarter	80%						
15	Direct labor per unit	0.80	hours					
16	Labor cost per hour	\$ 7.50						
17	Variable overhead	\$ 2.00	per hour of direct labor					
18	Fixed overhead	\$	60,000.00	\$ 60,000.00	\$ 60,000.00	\$ 60,000.00		
19	Depreciation included above	\$	12,000.00	\$ 12,000.00	\$ 12,000.00	\$ 12,000.00		
20	Variable sales cost	\$ 1.80						
21	Advertising	\$	-	\$ 20,000	\$ -	\$ 20,000		
22	Executive salaries	\$	50,000	\$ 50,000	\$ 50,000	\$ 50,000		
23	Insurance	\$	20,000	\$ 20,000	\$ 20,000	\$ 35,000		
24	Property taxes	\$		\$ 17,500	\$ 17,500	\$ 17,500		
25	Depreciation - selling	\$	10,000	\$ 10,000	\$ 10,000	\$ 10,000		
26	Equipment purchases	\$	30,000	\$ 45,000	\$ 22,500	\$ 18,000		
27	Dividends	\$	8,000	\$ 8,000	\$ 8,000	\$ 8,000		
28								

The red numbers in Figure 1 are the assumptions that students must change for their new product. As shown in Figure 2, they also need some information about the beginning balance sheet, borrowing, minimum cash level, the tax rate, and the amount of quarterly income taxes payments.

**Figure 2:
The Beginning Balance Sheet**

	A	B	C	D	E	F	H
28							
29	Beginning Balance Sheet						
30	Current Assets						
31	Cash					\$ 10,000	
32	Accounts receivable					\$ -	
33	Raw materials		0 Pounds		\$0.25	\$ -	
34	Finished products		0 Units		\$11.35	\$ -	\$ 10,000
35	Long-term assets						
36	Land					\$ -	
37	Plant and equipment				\$ -	\$ -	
38	Accumulated Depreciation				\$ -	\$ -	\$ -
39	Total assets						\$ 10,000
40							
41	Liabilities and Equity						
42	Liabilities						
43	Accounts payable					\$ -	
44	Bank Borrowing					\$ -	
45						\$ -	
46	Equity						
47	Common Stock					\$ 10,000	
48	Retained earnings					\$ -	
49						\$ 10,000	
50						\$ 10,000	
51	Minimum cash required	\$ 40,000.00					
52	Borrowing by multiples of	\$ 1,000.00					
53	Interest	10%					
54	Quarterly tax Payments	\$ 30,000.00					
55	Tax Rate	35%					

For accounting students, the same template is used, but they must review all formulas used and understand how to use them. Students are told to fill in the information from the example in the book that we are currently using for the graduate or undergraduate course (Garrison, Noreen, & Brewer, 2008) or (Hansen & Mowen, 2007).

How to use a template to teach the budget process

The first element of the budget process is to calculate revenues and cash inflows. Revenues are the product of quantity sold and the price per unit. Students are informed that there are many ways to determine the price, and that once the price is set the quantities can be obtained from the marketing department. Students must also understand that the cash might not be received at the same time as the sale.

One can determine, using cells B4:B5 in Figure 1, that because the company has a market share of 10% of a total market of 1,000,000 units, it is selling 100,000 units. Because of annual sales growth of 5%, the company is expecting sales of 105,000 next year. This number will be used with some calculations. However, this number is confusing for most non-accounting majors and the numbers are hidden. In the first quarter, the firm will have 10% of its sales which will be 10,000 units. The formula used in C8 is the following: $=B\$4*B\$5*C7$. This formula is copied to D8:F8.

The next step is to create the sales budget as shown in Figure 3:

Figure 3
The Sales Budget

	A	B	C	D	E	F	H
1			Year X Quarters				
2	Sales		1	2	3	4	Year
3	Budgeted sales in units		10,000	30,000	40,000	20,000	100,000
4	Selling price per unit		\$ 25.00	\$ 25.00	\$ 25.00	\$ 25.00	\$ 25.00
5	Sales		250,000	750,000	1,000,000	500,000	2,500,000

First quarter unit sales in C3 are determined by referencing the assumed amount with the formula: =Assumptions!C8. This formula is copied for each quarter, and the quantity for the year is the sum of the four quarters. In C4, the formula for selling price per unit is =Assumptions!\$B\$3. Finally, the formula for total sales is: =C3*C4.

From this table, the cash inflows can be calculated. As assumed in Assumptions!B9, seventy percent (70%) of first quarter sales will be collected in the first quarter (\$250,000 x 70%). The accounts receivable from the previous quarter will also be received in the first quarter. At the end of the fourth quarter, 30% of the sales of that quarter are still receivable. The information is summarized in the next table:

Figure 4
Cash Collections

	A	B	C	D	E	F	H
7	Cash Collections						
8	Accounts receivable		\$ -				\$ -
9	Same Quarter sales		\$ 175,000	\$ 525,000	\$ 700,000	\$ 350,000	\$ 1,750,000
10	Previous Quarter Sales		\$ 75,000	\$ 225,000	\$ 300,000	\$ 600,000	\$ 600,000
11	Cash Collections		\$ 175,000	\$ 600,000	\$ 925,000	\$ 650,000	\$ 2,350,000
12							
13	Accounts receivable at the end					\$ 150,000	\$ 150,000

Cell C8 is calculated as =Assumptions!F32. It represents the amount of accounts receivable at the beginning of the year, this would be the first amount received. Cell C9 is calculated by multiplying the amount of sales for the first quarter by the percentage of cash receipts the same quarter. The formula is =C5*Assumptions!\$B\$9. The remaining amount not received the same quarter will be received the following quarter. The formula in D10 is =C5-C9.

The next step is to look at the manufacturing process. The number of units to manufacture must be calculated. Because units must be fabricated before they are sold, we also have to calculate how many units are needed as a “buffer.” This is usually a percentage of the sales of the next quarter. The necessary information is in B10:B19, as shown in Figure 5:

Figure 5
Production Assumptions

	A	B	C	D	E	F
10	Ending inventory finished goods	25%	next quarter sales			
11	Raw material per unit	15	pounds per unit			
12	Raw material inventory	20%	next quarter purchases			
13	Cost of raw materials	\$ 0.25	per pound			
14	Cash payments same quarter	80%				
15	Direct labor per unit	0.80	hours			
16	Labor cost per hour	\$ 7.50				
17	Variable overhead	\$ 2.00	per hour of direct labor			
18	Fixed overhead		\$ 60,000.00	\$ 60,000.00	\$ 60,000.00	\$ 60,000.00
19	Depreciation included above		\$ 12,000.00	\$ 12,000.00	\$ 12,000.00	\$ 12,000.00

In the first quarter, the company needs not only to manufacture the 10,000 units, but also needs a buffer of 25% on the second quarter sales (30,000 x 25%), which means they need to manufacture a total of 17,500 units.

Figure 6 shows the production budget. The budgeted sales come from the sales budget (C3). The desired ending inventory is calculated with the formula: =D16*Assumptions!\$B\$10. This means that it is the percentage specified in B10 multiplied by the sales of next quarter. The beginning inventory in C19 is the amount in cell C34 in Figure 2, the number of units in the beginning inventory for the year. For all other quarters, the beginning inventory in a quarter is the ending inventory in the previous quarter.

Figure 6
The Production Budget

	A	B	C	D	E	F	H
15	Production budget						
16	Budgeted sales		10,000	30,000	40,000	20,000	100,000
17	Ending inventory		7,500	10,000	5,000	2,625	2,625
18	Total needs		17,500	40,000	45,000	22,625	102,625
19	Beginning Inventory		-	7,500	10,000	5,000	-
20	Production in units		17,500	32,500	35,000	17,625	102,625

Students are told that the security buffer for the fourth quarter was calculated as 2,625, which is 25% of 10,500 units. Students must remember that because growth was 5%, sales for the first quarter of next year would be 10,000 plus 5%, for a total of 10,500 units. Since this will be the required production for the first quarter of year 2, the company needs 25% of this amount at the end of quarter 4 of year 1. Students should notice that the ending inventory for the year (H17) is not the sum of quarterly ending inventory but the ending inventory at the end of the fourth quarter (F17). Similarly, the beginning inventory for the year (H19) is the beginning inventory for the first quarter (C19).

The first element of the manufacturing cost is the cost of raw materials. Since 15 pounds are needed for each unit, we would need 17,500 units at 15 pounds each for a total of 262,500 pounds of raw materials. However, raw materials have to be purchased in advance and a security “buffer” is also needed. In this case, it represents 20% of the purchase of the next quarter. I remind students to be careful about beginning and ending inventories. After the quantities have been determined, the cost is calculated by multiplying the pounds by the cost per pounds. The results are as follows:

Figure 7
The Materials Budget

	A	B	C	D	E	F	H
22 Direct Materials Budget							
23 Required Production			17,500	32,500	35,000	17,625	102,625
24 Raw Materials per unit			15.00	15.00	15.00	15.00	15.00
25 Raw materials needed			262,500	487,500	525,000	264,375	1,539,375
26 Ending inventory			97,500	105,000	52,875	47,250	
27 Total needs			360,000	592,500	577,875	311,625	1,539,375
28 Beginning Inventory			-	97,500	105,000	52,875	-
29 Purchases			360,000	495,000	472,875	258,750	1,586,625
30 Cost per pound			\$ 0.25	\$ 0.25	\$ 0.25	\$ 0.25	\$ 0.25
31 Cost of raw materials			\$ 90,000	\$ 123,750	\$ 118,219	\$ 64,688	\$ 396,656

The calculations for the materials budget are summarized in Figure 8:

Figure 8
Formulas for the Materials Budget

	A	C	D	E	F	H
22 Direct Materials Budget						
23 Required Production	=C20	=D20	=E20	=F20	=H20	
24 Raw Materials per unit	=Assumptions!\$B\$11	=Assumptions!\$B\$11	=Assumptions!\$B\$11	=Assumptions!\$B\$11	=Assumptions!\$B\$11	=Assumptions!\$B\$11
25 Raw materials needed	=C24*C23	=D24*D23	=E24*E23	=F24*F23	=H24*H23	
26 Ending inventory	=D25*Assumptions!\$B\$12	=E25*Assumptions!\$B\$12	=F25*Assumptions!\$B\$12	=G25*Assumptions!\$B\$12		
27 Total needs	=SUM(C25:C26)	=SUM(D25:D26)	=SUM(E25:E26)	=SUM(F25:F26)	=SUM(H25:H26)	
28 Beginning Inventory	=Assumptions!C33	=C26	=D26	=E26	=C28	
29 Purchases	=C27-C28	=D27-D28	=E27-E28	=F27-F28	=SUM(C29:F29)	
30 Cost per pound	=Assumptions!\$B\$13	=Assumptions!\$B\$13	=Assumptions!\$B\$13	=Assumptions!\$B\$13	=Assumptions!\$B\$13	=Assumptions!\$B\$13
31 Cost of raw materials	=C29*C30	=D29*D30	=E29*E30	=F29*F30	=H29*H30	

Because the cash outflow for each semester is only 80% (as specified in Assumptions!B14) of the purchases of the quarter, outflows need to be calculated in Figure 9.

Figure 9
Cash Outflows

	A	B	C	D	E	F	H
33 Cash output							
34 Accounts payable beginning			\$ -				\$ -
35 Same Quarter			\$ 72,000	\$ 99,000	\$ 94,575	\$ 51,750	\$ 317,325
36 Next Quarter			\$ 18,000	\$ 24,750	\$ 24,750	\$ 23,644	\$ 66,394
37 Cash Output			\$ 72,000	\$ 117,000	\$ 119,325	\$ 75,394	\$ 383,719
38							
39 Account payable at the end						\$ 12,938	\$ 12,938

Students should be told to notice that the formulas used here are very similar to the formulas used in for the cash inflows (see Figure 4).

The other elements of the manufacturing process are easier to determine than the direct materials cost because neither direct labor nor overhead need a buffer. Unlike raw materials, you cannot buy labor in advance and store it to be used later. The calculations for direct labor and overhead are straightforward and are shown in Figure 10. Students are also reminded that because depreciation is not a cash outflow, it has to be taken out of the expenses to determine the cash outflows.

Figure 10
The Direct Labor and Overhead Budgets

	A	B	C	D	E	F	H
41 Direct Labor Budget							
42 Required Production			17,500	32,500	35,000	17,625	102,625
43 Direct labor time per unit			0.80	0.80	0.80	0.80	0.80
44 Total hours needed			14,000	26,000	28,000	14,100	82,100
45 Labor cost per hour			\$ 7.50	\$ 7.50	\$ 7.50	\$ 7.50	\$ 7.50
46 Total Labor Cost			\$ 105,000	\$ 195,000	\$ 210,000	\$ 105,750	\$ 615,750
47 Pay as you go							
48							
49 Overhead Budget							
50 Direct Labor Hours			14,000	26,000	28,000	14,100	82,100
51 Variable Overhead Rate			\$ 2.00	\$ 2.00	\$ 2.00	\$ 2.00	\$ 2.00
52 Variable Overhead			\$ 28,000	\$ 52,000	\$ 56,000	\$ 28,200	\$ 164,200
53 Fixed Overhead			\$ 60,000	\$ 60,000	\$ 60,000	\$ 60,000	\$ 240,000
54 Total Overhead			\$ 88,000	\$ 112,000	\$ 116,000	\$ 88,200	\$ 404,200
55 Less depreciation included above			\$ 12,000	\$ 12,000	\$ 12,000	\$ 12,000	\$ 48,000
56 Total Overhead Disbursement			\$ 76,000	\$ 100,000	\$ 104,000	\$ 76,200	\$ 356,200

The numbers in Figure 10 were calculated using the formulas shown in Figure 11:

Figure 11
Formulas for the Labor and Overhead Budgets

	A	B	C	D	E	F	H
41 Direct Labor Budget							
42 Required Production		=C20	=D20	=E20	=F20	=H20	
43 Direct labor time per unit		=Assumptions!\$B\$15	=Assumptions!\$B\$15	=Assumptions!\$B\$15	=Assumptions!\$B\$15	=Assumptions!\$B\$15	=Assumptions!\$B\$15
44 Total hours needed		=C43*C42	=D43*D42	=E43*E42	=F43*F42	=H43*H42	
45 Labor cost per hour		=Assumptions!\$B\$16	=Assumptions!\$B\$16	=Assumptions!\$B\$16	=Assumptions!\$B\$16	=Assumptions!\$B\$16	=Assumptions!\$B\$16
46 Total Labor Cost		=C44*C45	=D44*D45	=E44*E45	=F44*F45	=H44*H45	
47 Pay as you go							
48							
49 Overhead Budget							
50 Direct Labor Hours		=C44	=D44	=E44	=F44	=H44	
51 Variable Overhead Rate		=Assumptions!\$B\$17	=Assumptions!\$B\$17	=Assumptions!\$B\$17	=Assumptions!\$B\$17	=Assumptions!\$B\$17	=Assumptions!\$B\$17
52 Variable Overhead		=C51*C50	=D51*D50	=E51*E50	=F51*F50	=H51*H50	
53 Fixed Overhead		=Assumptions!C18	=Assumptions!D18	=Assumptions!E18	=Assumptions!F18	=SUM(C53:F53)	
54 Total Overhead		=C52+C53	=D52+D53	=E52+E53	=F52+F53	=H52+H53	
55 Less depreciation included above		=Assumptions!C19	=Assumptions!D19	=Assumptions!E19	=Assumptions!F19	=SUM(C55:F55)	
56 Total Overhead Disbursement		=C54-C55	=D54-D55	=E54-E55	=F54-F55	=H54-H55	

The next step in the budget process is to add the selling and administrative expenses. For this step, students must remember that variable selling costs are per unit sold, not per unit manufactured. The other selling and administrative costs are a fixed determined amount per quarter and are easy to calculate. Again, we have to add back depreciation to determine the cash outflow.

Figure 12
The Selling and Administrative Expense Budget

	A	B	C	D	E	F	H
58 Selling and Administrative Expenses							
59 Budgeted sales			10,000	30,000	40,000	20,000	100,000
60 Variable costs per unit			\$ 1.80	\$ 1.80	\$ 1.80	\$ 1.80	\$ 1.80
61 Total Variable costs			\$18,000.00	\$54,000.00	\$72,000.00	\$36,000.00	\$ 180,000
62 Fixed costs							
63 Advertising			\$ -	\$ 20,000	\$ -	\$ 20,000	\$ 40,000
64 Executive salaries			\$ 50,000	\$ 50,000	\$ 50,000	\$ 50,000	\$ 200,000
65 Insurance			\$ -	\$ 20,000	\$ -	\$ 35,000	\$ 55,000
66 Property taxes			\$ -	\$ -	\$ -	\$ 17,500	\$ 17,500
67 Depreciation - selling			\$ 10,000	\$ 10,000	\$ 10,000	\$ 10,000	\$ 40,000
68 Total fixed			\$ 60,000	\$ 100,000	\$ 60,000	\$ 132,500	\$ 352,500
69 Total selling and administrative			\$ 78,000	\$ 154,000	\$ 132,000	\$ 168,500	\$ 532,500
70 Less depreciation			\$ 10,000	\$ 10,000	\$ 10,000	\$ 10,000	\$ 40,000
71 Selling & Administrative Disbursement			\$ 68,000	\$ 144,000	\$ 122,000	\$ 158,500	\$ 492,500

As shown in Figure 13, the variable costs are calculated while the fixed costs are just taken from the assumption section.

Figure 13
Formulas for the Selling and Administrative Expense Budget

	A	B	C	D	E	F	H
58 Selling and Administrative Expenses							
59 Budgeted sales			=C16	=D16	=E16	=F16	=H16
60 Variable costs per unit			=Assumptions!\$B\$20	=Assumptions!\$B\$20	=Assumptions!\$B\$20	=Assumptions!\$B\$20	=Assumptions!\$B\$20
61 Total Variable costs			=C59*C60	=D59*D60	=E59*E60	=F59*F60	=H59*H60
62 Fixed costs							
63 Advertising			=Assumptions!C21	=Assumptions!D21	=Assumptions!E21	=Assumptions!F21	=SUM(C63:F63)
64 Executive salaries			=Assumptions!C22	=Assumptions!D22	=Assumptions!E22	=Assumptions!F22	=SUM(C64:F64)
65 Insurance			=Assumptions!C23	=Assumptions!D23	=Assumptions!E23	=Assumptions!F23	=SUM(C65:F65)
66 Property taxes			=Assumptions!C24	=Assumptions!D24	=Assumptions!E24	=Assumptions!F24	=SUM(C66:F66)
67 Depreciation - selling			=Assumptions!C25	=Assumptions!D25	=Assumptions!E25	=Assumptions!F25	=SUM(C67:F67)
68 Total fixed			=SUM(C63:C67)	=SUM(D63:D67)	=SUM(E63:E67)	=SUM(F63:F67)	=SUM(H63:H67)
69 Total selling and administrative			=C68+C61	=D68+D61	=E68+E61	=F68+F61	=H68+H61
70 Less depreciation			=C67	=D67	=E67	=F67	=H67
71 Selling & Administrative Disbursement			=C69-C70	=D69-D70	=E69-E70	=F69-F70	=H69-H70

Creating the Cash Budget

Now is the time to calculate the Cash Budget (see the Financials worksheet). The first element is to look at the beginning balance of the cash and add the cash inflows from the sales budget. The beginning cash plus cash receipts is the total available.

The next step is to look at all of the cash disbursements from purchases, the direct labor budget, the overhead budget and the selling and administrative budget. Two more disbursements are also taken into account: equipment purchases and dividends (see C26:F27 in Figure 1). These are added to the disbursements. We also add one more line for interest that we will need to calculate.

Figure 14
The Cash Budget

	A	B	C	D	E	F
1	Cash Flow Budget					
2		Year X Quarters				
3		1	2	3	4	Year
4	Cash Beginning	\$ 10,000	\$ 40,000	\$ 40,650	\$ 40,138	\$ 10,000
5	Cash receipts	\$ 175,000	\$ 600,000	\$ 925,000	\$ 650,000	\$ 2,350,000
6	Total available	\$ 185,000	\$ 640,000	\$ 965,650	\$ 690,138	\$ 2,360,000
7						
8	Cash disbursement					
9	Purchases	\$ 72,000	\$ 117,000	\$ 119,325	\$ 75,394	\$ 383,719
10	Direct labor	\$ 105,000	\$ 195,000	\$ 210,000	\$ 105,750	\$ 615,750
11	Overhead	\$ 76,000	\$ 100,000	\$ 104,000	\$ 76,200	\$ 356,200
12	Selling and Administrative	\$ 68,000	\$ 144,000	\$ 122,000	\$ 158,500	\$ 492,500
13	Equipment purchases	\$ 30,000	\$ 45,000	\$ 22,500	\$ 18,000	\$ 115,500
14	Dividends	\$ 8,000	\$ 8,000	\$ 8,000	\$ 8,000	\$ 32,000
15	Interest payments		\$ 21,350	\$ 26,688	\$ 26,688	\$ 74,725
16	Estimated Income Taxes Payments	\$ 30,000	\$ 30,000	\$ 30,000	\$ 30,000	\$ 120,000
17	Total disbursements	\$ 389,000	\$ 660,350	\$ 642,513	\$ 498,531	\$ 2,190,394
18	Excess (Deficit) cash	\$ (204,000)	\$ (20,350)	\$ 323,138	\$ 191,606	\$ 169,606

The difference between the total available and the total disbursements will determine the excess (or deficit) of cash in row 18. All numbers used in the cash budget come from numbers previously calculated .

We now need to determine the amount to borrow and when it will be reimbursed. Two important assumptions are made here: Borrowing is always done on the first day of a quarter and repayment is always made on the last day of a quarter. Additionally, we make another assumption about the amount borrowed and how to round it up. Finally an assumption is made about the tax rate and the quarterly payments made throughout the year.

Figure 15
Borrowing and Taxes Assumptions

	A	B
51	Minimum cash required	\$40,000.00
52	Borrowing by multiples of	\$ 1,000.00
53	Interest	10%
54	Quarterly tax Payments	\$30,000.00
55	Tax Rate	35%

In the example above, we have a deficit of \$204,000 and since we need a minimum of \$40,000 (B51) in cash, we must borrow \$244,000, which is a multiple of \$1,000 (B52). The amount is calculated the following way:

$$=IF(B18<Assumptions!B51,ROUNDUP(-B18+Assumptions!B51,-3),0)$$

This means that if the excess or deficit of cash is less than the amount specified in the assumptions, you round up the sum of the deficit cash plus the desired amount. The roundup is to the third number to the left of the decimal, which means rounded up to the next thousand. This amount will appear on the line labeled Borrowings in Figure 16 (C20):

Figure 16
Borrowing and Interest Payments

	A	B	C	D	E	F
15 Interest payments			\$ 21,350	\$ 26,688	\$ 26,688	\$ 74,725
16 Estimated Income Taxes Payments		\$ 30,000	\$ 30,000	\$ 30,000	\$ 30,000	\$ 120,000
17 Total disbursements		\$ 389,000	\$ 660,350	\$ 642,513	\$ 498,531	\$ 2,190,394
18 Excess (Deficit) cash		\$ (204,000)	\$ (20,350)	\$ 323,138	\$ 191,606	\$ 169,606
19 Financing						
20 Borrowing (first day)		\$ 244,000	\$ 61,000	\$ -	\$ -	\$ 305,000
21 Repayment (last day)			\$ -	\$ (283,000)	\$ (22,000)	\$ (305,000)
22 Adjustment to repayment				\$ -	\$ -	\$ -
23 Accrued Interest		\$ 21,350	\$ 26,688	\$ 26,688	\$ 1,925	\$ 76,650
24 Interest					\$ -	\$ -
25 Adjustment to interest				\$ -	\$ -	\$ -
26 Total financing		\$ 244,000	\$ 61,000	\$ (283,000)	\$ (22,000)	\$ -
27						
28 Cash at the end		\$ 40,000	\$ 40,650	\$ 40,138	\$ 169,606	\$ 169,606

Accrued interest is next calculated in B23 using the formula =B20*Assumptions!B55/4.

The interest is thus calculated as \$21,350 which will be paid the first day of the next quarter to avoid compound interest. This was indicated as line 15 of the disbursement of the next quarter (C15). Students are also told to note that cash at the end of the first quarter (B28) will be the cash at the beginning of the second quarter (C4 in Figure 14).

The last step of the budget process is to prepare the annual financial statements. The manufacturing cost and the income statement use revenues and expenses, not the inflows or outflows of cash. This is the Manufacturing Costs Statement is shown in Figure 17:

Figure 17
The Manufacturing Cost Statement

	A	B	C	D	E	F
31 Manufacturing Costs						
32 Direct Materials						
33		Beginning Inventory		\$ -		
34		Plus purchases		\$ 396,656		
35		Less ending inventory		\$ (11,813)	\$ 384,844	
36 Direct Labor					\$ 615,750	
37 Overhead					\$ 404,200	
38 Total Manufacturing Costs					\$ 1,404,794	

The Manufacturing Costs Statement is used to determine how much it cost to produce the goods manufactured (or finished) during the year.

The Income Statement is shown next.

Figure 18
The Income Statement

	A	B	C	D	E	F
42	Income Statement					
43						
44	Sales					\$ 2,500,000
45	Cost of Goods Sold	Beginning Inventory		\$ -		
46		Plus Manufacturing Costs		\$ 1,404,794		
47		Less Ending Inventory		\$ (29,794)		
48	Cost of Goods Sold					\$ 1,375,000
49	Gross margin					\$ 1,125,000
50	Selling and Administrative					\$ 532,500
51	Net Income before interest					\$ 592,500
52	Interest Expense					\$ 76,650
53	Net Income before taxes					\$ 515,850
54	Taxes					\$ 180,548
55	Net Income					\$ 335,303

In order to determine the Cost of Goods Sold in Figure 18, we need to add the beginning inventory of finished goods plus the manufacturing cost and subtract the ending inventory of finished goods. The tax expense is calculated by multiplying the net income before taxes (515,850) by the assumed tax rate (see B55 in Figure 15). The formula used is: =F53*Assumptions!B55 Finally, the balance sheet is prepared.

Figure 19
The Balance Sheet

	A	B	C	D	E	F
57	Balance Sheet					
58						
59	Current Assets					
60	Cash				\$ 169,606	
61	Accounts receivable				\$ 150,000	
62	Raw materials				\$ 11,813	
63	Finished products				\$ 29,794	\$ 361,213
64	Long-term assets					
65	Land				\$ -	
66	Plant and equipment			\$ 115,500		
67	Accumulated Depreciation			\$ (88,000)	\$ 27,500	\$ 27,500
68						
69	Total assets					\$ 388,713
70						
71	Liabilities and Equity					
72	Liabilities					
73	Accounts payable					\$ 12,938
74	Taxes payable					\$ 60,548
75	Bank Borrowing and accrued interests					\$ 1,925
76	Total Liabilities					\$ 75,411
77						
78	Equity					
79	Common Stock					\$ 10,000
80	Beginning Retained earnings			\$ -		
81	Net income			\$ 335,303		
82	Less Dividends			\$ (32,000)		
83	Retained earnings at the end					\$ 303,303
84	Total Equity					\$ 313,303
85						
86	Total Liabilities and Equity					\$ 388,713

The total of the Assets is equal to the total of Liabilities and Stockholders' Equity. Students can use the financial statements to do a ratio analysis or any other form of financial analysis. There is an amount to pay for income taxes because the amount paid (\$120,000) as per line 54 of Figure

15, was not enough to cover the real income tax expense (\$180,548 as per F54 in Figure 18) shown in the income statement. Students are told that every quarter a company must make quarterly payments to the IRS to cover its tax obligation. The firm estimates the quarterly payments at the beginning of the year, usually based on the tax expense from the year before. At the end of the year, income taxes are calculated, and the difference will result in an amount payable or receivable. For this template, we just apply the tax rate assumed (see B55 in Figure 15) to the net income to calculate the tax expense.

Comment [A1]: I don't think that I understand this part. It seems out of place. Maybe it should be mentioned below figure 18 (the income statement)? I also don't quite understand the relationship of this calculation to the \$30,000 that was assumed in the budget. I don't see where that \$30,000 was ever used, and it isn't referenced in the formula. So, why was that assumed in the first place and how should it affect this calculation?

Explaining the process and doing the whole budget usually takes about three hours.

How to use the template to do a scenario analysis.

Imagine that your sales force told you that if you reduce the price by \$1 you could increase market share by 2%. Should you do it? It could take another 3 hours to answer if done by hand. This is when the spreadsheet is especially useful. Change the two assumptions and see the results immediately. Go to the Assumptions worksheet and change B3 from \$25 to \$24. Also change B5 from 10% to 12% and see the results in the Financials Sheet.

Figure 20
New Assumptions

	A	B
1	Fill the Red Numbers with your information	
2		
3	Selling price per unit	\$ 24.00
4	Total market per year	1,000,000
5	Market share	12%

Net Income is now \$410,215 an improvement from the situation before. Note that every number of the income statement was recalculated automatically.

Figure 21
The New Income Statement

	A	B	C	D	E	F
42	Income Statement					
43						
44	Sales					\$ 2,880,000
45	COGS	Beginning inventory			\$ -	
46		Plus Manufacturing Costs			\$ 1,637,753	
47		Less Ending Inventory			\$ (35,753)	
48	COGS					\$ 1,602,000
49	Gross margin					\$ 1,278,000
50	Selling and Administrative					\$ 568,500
51	Net Income before interest					\$ 709,500
52	Interest Expense					\$ 78,400
53	Net Income before taxes					\$ 631,100
54	Taxes					\$ 220,885
55	Net Income					\$ 410,215

Also, since your financials are all there, you could calculate the impact on your ratios. You immediately get the results of any change in your assumptions. Any number in red can be changed and you will see the results immediately.

Conclusion

By using a template with an assumptions sheet, you can very rapidly see the impact of a decision. This is very useful for both accountants and non-accountants. The advantage of a template is that the user does not have to know how to calculate all of the elements of the budget and can concentrate on the results.

Accounting majors should be shown step-by-step how they can construct such a template for their own needs. They are told that it is important to write the formulas correctly and to make a reference to a cell address (B3) instead of just using the amount \$25.

Students are also told that the approach used here is more realistic than the approach used in many books by looking at the total market and a specific market share for their business instead of just using quantities as most accounting books do.

It takes a lot more time to build the template but once it is built, people will save hundreds of hours every time they have to do a budget and every time they have to answer a simple question such as what happens if my raw materials cost is increased by 10%. With a flexible budget process, it is very easy to see the impact of some changes in a few key strokes instead of taking another three hours to recalculate everything.

References

Garrison, R., Noreen, E., & Brewer, P. (2008). *Managerial Accounting* (12th ed.). New York, NY: McGraw-Hill Irwin.

Hansen, D. R., & Mowen, M. M. (2007). *Managerial Accounting* (8th ed.). Mason, OH: Thomson South-Western.

Kuratko, D. F. (2009). *Entrepreneurship: Theory, Process, Practice* (8th ed.). Mason, OH: South-Western Cengage Learning.

The Entrepreneurship Foundation. (2009). *Business Planning for New Ventures*. Retrieved December 20, 2009, from <http://www.entrepreneurshipfoundation.org/resources.html>